

Software

Program, Application

Programming Language

Compiler/Interpreter

Operating System

Instruction Set Architecture



Microarchitecture

Digital Logic

Devices (transistors, etc.)

Solid-State Physics

next
few
weeks



Hardware

CSAPP book is **very useful** and well-aligned with class for the remainder of the course.

C to Machine Code and x86 Basics

ISA context and x86 history

Translation tools: C --> assembly <--> machine code

x86 Basics:

- Registers

- Data movement instructions

- Memory addressing modes


- Arithmetic instructions

Turning C into Machine Code

C Code

```
void sumstore(long x, long y,  
              long *dest) {  
    long t = x + y;  
    *dest = t;  
}
```

sum.c




Generated x86 Assembly Code

Human-readable language close to machine code.

```
sum:  
    addq %rdi,%rsi  
    movq %rsi,(%rdx)  
    retq
```


sum.s



Object Code

```
01010101100010011110010110  
00101101000101000011000000  
00110100010100001000100010  
01111011000101110111000011
```

sum.o



Executable: sum

Resolve references between object files,
libraries, (re)locate data

Machine Instruction Example

```
*dest = t;
```

```
movq %rsi, (%rdx)
```

```
0x400539: 48 89 32
```

C Code

Store value `t` where indicated by `dest`

Assembly

Move 8-byte value to memory

"Quad word" in x86-64 speak

Operands:

`t`: Register `%rsi`

`dest`: Register `%rdx`

`*dest`: Memory **M**[`%rdx`]

Object Code

3-byte instruction encoding


Stored at address `0x400539`

Disassembling Object Code

```
001011010001010000110000
001101000101000010001000
011110110001011101110000
...
```

Disassembled by `objdump -d sum`

```
0000000000400536 <sumstore>:
 400536: 48 01 fe add %rdi,%rsi
 400539: 48 89 32 mov %rsi,(%rdx)
 40053c: c3      retq
```



Object

```
0x00400536:
0x48
0x01
0xfe
0x48
0x89
0x32
0xc3
```

Disassembled by GDB

```
0x0000000000400536 <+0>: add %rdi,%rsi
0x0000000000400539 <+3>: mov %rsi,(%rdx)
0x000000000040053c <+6>: retq
```

`$ gdb sum`

`(gdb) disassemble sumstore`

(disassemble function)

`(gdb) x/7b sum`

(examine the 13 bytes starting at sum)

CISC vs. RISC

x86: real ISA, widespread

CISC: maximalism

Complex Instruction Set Computer

Many instructions, specialized.

Variable-size encoding,
complex/slow decode.

Gradual accumulation over time.

Original goal:

- humans program in assembly
- or simple compilers generate assembly by template
- hardware supports many patterns as single instructions
- fewer instructions per SLOC

Usually fewer registers.

We will stick to a small subset.

HW: toy, but based on real MIPS ISA

RISC: minimalism

Reduced Instruction Set Computer

Few instructions, general.

Regular encoding,
simple/fast decode.

1980s+ reaction to bloated ISAs.

Original goal:

- humans use high-level languages
- smart compilers generate highly optimized assembly
- hardware supports fast basic instructions
- more instructions per SLOC

Usually many registers.

a brief history of x86

Word
Size

ISA

First

Year

16

8086

Intel 8086

1978

First 16-bit processor. Basis for IBM PC & DOS
1MB address space

32

IA32

Intel 386

1985

First 32-bit ISA.
Flat addressing, improved OS support

2016: most laptops,
desktops, servers.

240 now:


64

x86-64

AMD Opteron 2003*

Slow AMD/Intel conversion, slow adoption.
*Not actually x86-64 until few years later.
Mainstream only after ~10 years.

ISA View



x86-64 registers

<code>%rax</code>	Return Value
<code>%rbx</code>	
<code>%rcx</code>	Function Argument 4
<code>%rdx</code>	Function Argument 3
<code>%rsi</code>	Function Argument 2
<code>%rdi</code>	Function Argument 1
<code>%rsp</code>	Special purpose: Stack pointer
<code>%rbp</code>	

64-bits wide

<code>%r8</code>	Function Argument 5
<code>%r9</code>	Function Argument 6
<code>%r10</code>	
<code>%r11</code>	
<code>%r12</code>	
<code>%r13</code>	
<code>%r14</code>	
<code>%r15</code>	

Some have special uses
for particular instructions


x86 historical artifacts

Low 32 bits of 64-bit register

<code>%rax</code>	<code>%eax</code>
<code>%rbx</code>	<code>%ebx</code>
<code>%rcx</code>	<code>%ecx</code>
<code>%rdx</code>	<code>%edx</code>
<code>%rsi</code>	<code>%esi</code>
<code>%rdi</code>	<code>%edi</code>
<code>%rsp</code>	<code>%esp</code>
<code>%rbp</code>	<code>%ebp</code>

<code>%r8</code>	<code>%r8d</code>
<code>%r9</code>	<code>%r9d</code>
<code>%r10</code>	<code>%r10d</code>
<code>%r11</code>	<code>%r11d</code>
<code>%r12</code>	<code>%r12d</code>
<code>%r13</code>	<code>%r13d</code>
<code>%r14</code>	<code>%r14d</code>
<code>%r15</code>	<code>%r15d</code>

x86 historical artifacts



x86: Three Basic Kinds of Instructions

1. Data movement between memory and register

Load data from memory into register

$\%reg \leftarrow \text{Mem}[\text{address}]$

Store register data into memory

$\text{Mem}[\text{address}] \leftarrow \%reg$

Memory is an
array[] of bytes!

2. Arithmetic/logic on register or memory data

$c = a + b;$

$z = x \ll y;$

$i = h \& g;$

3. Comparisons and Control flow to choose next instruction

Unconditional jumps to/from procedures

Conditional branches

Data movement instructions

mov *Source, Dest*

data size is one of {**b**, **w**, **l**, **q**}

movq *Source, Dest*:

Move 8-byte “quad word”

movl *Source, Dest*:

Move 4-byte “long word”


movw *Source, Dest*:

Move 2-byte “word”

movb *Source, Dest*:

Move 1-byte “byte”

← Historical terms based on the 16-bit days,
not the current machine word size (64 bits)



Data movement instructions

`movq Source, Dest:`

Operand Types:



Immediate: Literal integer data

Examples: `$0x400`, `$-533`

Register: One of 16 registers

Examples: `%rax`, `%rdx`

Memory: 8 consecutive bytes in memory, at address held by register

Simplest example: `(%rax)`

mov Operand Combinations

	Source	Dest	Src, Dest	C Analog
movq	Imm	Reg	movq \$0x4, %rax	a = 0x4;
		Mem	movq \$-147, (%rax)	*p = -147;
	Reg	Reg	movq %rax, %rdx	d = a;
		Mem	movq %rax, (%rdx)	*q = a;
	Mem	Reg	movq (%rax), %rdx	d = *p;

Cannot do memory-memory transfer with a single instruction.

How would you do it?

Basic Memory Addressing Modes

Indirect **(R)** **Mem[Reg[R]]**

Register R specifies the memory address

```
movq (%rcx) , %rax
```

Displacement **D(R)** **Mem[Reg[R]+D]**

Register R specifies base memory address

(e.g. the start of an object)

Displacement (offset) D specifies offset from base address

(e.g. a field in the object)


```
movq 8(%rsp) , %rdx
```


Pointers and Memory Addressing

```
void swap(long* xp, long* yp) {  
    long t0 = *xp;  
    long t1 = *yp;  
    *xp = t1;  
    *yp = t0;  
}
```

```
swap:  
    movq (%rdi), %rax  
    movq (%rsi), %rdx  
    movq %rdx, (%rdi)  
    movq %rax, (%rsi)  
    retq
```

<u>Register</u>		<u>Variable</u>
%rdi	↔	xp
%rsi	↔	yp
%rax	↔	t0
%rdx	↔	t1



Complete Memory Addressing Modes

General Form:

$$\mathbf{D(Rb,Ri,S)} \quad \text{Mem}[\text{Reg}[Rb] + S * \text{Reg}[Ri] + D]$$

D: Literal “displacement” value represented in 1, 2, or 4 bytes

Rb: Base register: Any register

Ri: Index register: Any except `%rsp`

S: Scale: 1, 2, 4, or 8 (*why these numbers?*)

Special Cases:

Implicitly:

(Rb,Ri) Mem[Reg[Rb]+Reg[Ri]] (S=1,D=0)

D(Rb,Ri) Mem[Reg[Rb]+Reg[Ri]+D] (S=1)

(Rb,Ri,S) Mem[Reg[Rb]+S*Reg[Ri]] (D=0)

Address Computation Examples



Register contents

<code>%rdx</code>	<code>0xf000</code>
<code>%rcx</code>	<code>0x100</code>

Addressing modes

$$D(Rb, Ri, S) \text{ Mem}[Reg[Rb] + S * Reg[Ri] + D]$$

Address Expression	Address Computation	Address
<code>0x8(%rdx)</code>		
<code>(%rdx, %rcx)</code>		
<code>(%rdx, %rcx, 4)</code>		
<code>0x80(, %rdx, 2)</code>		

"Lovely Efficient Arithmetic"

leaq Src, Dest *load effective address*

Src is address mode expression

Store **address computed** in Dest

Example: `leaq (%rdx,%rcx,4), %rax`

DOES NOT ACCESS MEMORY



Uses

`p = &x[i];`

Arithmetic of the form $x + k*i$

$k = 1, 2, 4, \text{ or } 8$

leaq vs. movq example

Registers


%rax	
%rbx	
%rcx	0x4
%rdx	0x100
%rdi	
%rsi	

Memory

	Address
0x400	0x120
0xf	0x118
0x8	0x110
0x10	0x108
0x1	0x100

```
leaq (%rdx,%rcx,4), %rax
movq (%rdx,%rcx,4), %rbx
leaq (%rdx), %rdi
movq (%rdx), %rsi
```


Memory Layout



Call Stack

Memory region for temporary storage managed with stack discipline.


`%rsp` holds lowest stack address (address of "top" element)



Call Stack: Push

`pushq Src`


1. Fetch value from *Src*
2. Decrement `%rsp` by 8 (*why 8?*)
3. Store value at new address given by `%rsp`



Call Stack: Pop

popq *Dest*

1. Load value from address `%rsp`
2. Write value to *Dest*
3. Increment `%rsp` by 8



Those bits are still there;
we're just not using them.

Procedure Preview (more soon)

call, ret, push, pop

Procedure arguments passed in 6 registers:


%rax	Return Value	%r8	Argument 5
%rbx		%r9	Argument 6
%rcx	Argument 4	%r10	
%rdx	Argument 3	%r11	
%rsi	Argument 2	%r12	
%rdi	Argument 1	%r13	
%rsp	Stack pointer	%r14	
%rbp		%r15	

Return value in %rax.

Allocate/push new *stack frame* for each procedure call.

Some local variables,
saved register values, extra arguments

Deallocate/pop frame before return.



Arithmetic Operations

Two-operand instructions:

Format

`addq Src, Dest`

`subq Src, Dest`

`imulq Src, Dest`

`shlq Src, Dest`

`sarq Src, Dest`

`shrq Src, Dest`

`xorq Src, Dest`

`andq Src, Dest`

`orq Src, Dest`

Computation

$Dest = Dest + Src$

$Dest = Dest - Src$

$Dest = Dest * Src$

$Dest = Dest \ll Src$

$Dest = Dest \gg Src$

$Dest = Dest \gg Src$

$Dest = Dest \wedge Src$

$Dest = Dest \& Src$

$Dest = Dest | Src$



argument order

a.k.a salq

Arithmetic

Logical

One-operand (unary) instructions

`incq Dest`

$Dest = Dest + 1$

increment

`decq Dest`

$Dest = Dest - 1$

decrement

`negq Dest`

$Dest = -Dest$

negate

`notq Dest`

$Dest = \sim Dest$

bitwise complement

See CSAPP 3.5.5 for: `mulq`, `cqto`, `idivq`, `divq`

leaq for arithmetic

```
long arith(long x, long y,  
           long z) {  
    long t1 = x+y;  
    long t2 = z+t1;  
    long t3 = x+4;  
    long t4 = y * 48;  
    long t5 = t3 + t4;  
    long rval = t2 * t5;  
    return rval;  
}
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rdx	Argument z
%rax	
%rcx	

```
arith:  
    leaq    (%rdi,%rsi), %rax  
    addq    %rdx, %rax  
    leaq    (%rsi,%rsi,2), %rdx  
    salq    $4, %rdx  
    leaq    4(%rdi,%rdx), %rcx  
    imulq   %rcx, %rax  
    ret
```

Another example

```
long logical(long x, long y) {
    long t1 = x^y;
    long t2 = t1 >> 17;
    long mask = (1<<13) - 7;
    long rval = t2 & mask;
    return rval;
}
```

Register	Use(s)
%rdi	Argument x
%rsi	Argument y
%rax	

logical:

```
movq %rdi,%rax
xorq %rsi,%rax
sarq $17,%rax
andq $8185,%rax
retq
```